- **Calculating directional derivatives in polar coordinates isn't standard. It is an interesting exercise in differential geometry, though. See http://www.shef.ac.uk/physics/teaching/phy314/relativity/previous_years/lectures/condensed/lec_6_relativity_comps_summary.pdf**

In[2]:=
```
x[r_, θ_] := r Cos[θ]
y[r_, θ_] := r Sin[θ]
polar = {r, θ};
xv = {x[r, θ], y[r, θ]};
dxdp[i_, μ_] := D[xv[[i]], polar[[μ]]]
g[μ_, ν_] := Sum[dxdp[i, μ] dxdp[i, ν], {i, 1, 2}] // Simplify
```

In[8]:=
```
Table[g[μ, ν], {μ, 1, 2}, {ν, 1, 2}] // MatrixForm
```

Out[8]//MatrixForm=

$$\begin{pmatrix} 1 & 0 \\ 0 & r^2 \end{pmatrix}$$

In[10]:=
```
gUpper[μ_, ν_] := {{1, 0}, {0, 1 / r^2}}[[μ]][[ν]]
Γ[α_][β_, γ_] := Sum[(gUpper[α, σ] / 2)
   (D[g[σ, β], polar[[γ]]] + D[g[σ, γ], polar[[β]]] - D[g[β, γ], polar[[σ]]]), {σ, 1, 2}]
```

In[12]:=
```
Table[Γ[α][β, γ], {α, 1, 2}, {β, 1, 2}, {γ, 1, 2}] // MatrixForm
```

Out[12]//MatrixForm=

$$\begin{pmatrix} \begin{pmatrix} 0 \\ 0 \end{pmatrix} & \begin{pmatrix} 0 \\ -r \end{pmatrix} \\ \begin{pmatrix} 0 \\ \frac{1}{r} \end{pmatrix} & \begin{pmatrix} \frac{1}{r} \\ 0 \end{pmatrix} \end{pmatrix}$$

- **Total derivative Dn^*μ*_*ν* = partial_*ν* n^*μ* + Γ^*μ*_{*νσ*} n^*σ***

In[13]:=
```
TotalDerivative[n_][μ_][ν_] := D[n[[μ]], polar[[ν]]] + Sum[Γ[μ][σ, ν] n[[σ]], {σ, 1, 2}]
```

- **Divergence div n = D_*μ* n^*μ***

In[23]:=
```
Div[n_] := Sum[TotalDerivative[n][μ][μ], {μ, 1, 2}]
```

In[24]:=
```
nVec = {nr[r, θ], nθ[r, θ]};
Div[nVec]
```

Out[25]= $\frac{\text{nr}[r,\theta]}{r} + \text{n}\theta^{(0,1)}[r,\theta] + \text{nr}^{(1,0)}[r,\theta]$

- **This isn't the traditional formula, This is because of two different conventions for writing vector fields: that for differential geometry {nr, n*θ*} and that for polar coordinates {Nr, N*θ*}.
In differential geometry, vectors are written in terms of the local basis for the tangent space *δ*r, *δθ*, given by raising the index on the gradients of the two coordinates. In polar coordinates, vectors are written in terms of normalized vectors hat r and hat *θ*. The problem is that *δθ* isn't normalized; it's of length r, so n*θ* = N*θ*/r:**

In[26]:=
```
NVec = {Nr[r, θ], Nθ[r, θ] / r};
 Div[NVec]
```

Out[27]= $\frac{\text{Nr}[r,\theta]}{r} + \frac{\text{N}\theta^{(0,1)}[r,\theta]}{r} + \text{Nr}^{(1,0)}[r,\theta]$

- ### Now we can investigate the directional derivative

In[28]:=
```
nDotGradn[n_] := Table[Sum[n[[α]] TotalDerivative[n][β][α], {α, 1, 2}], {β, 1, 2}] // Simplify
```

In[29]:=
```
nDotGradn[NVec]
```

Out[29]=
$$\left\{\frac{N\theta[r,\theta]\left(-N\theta[r,\theta]+Nr^{(0,1)}[r,\theta]\right)}{r}+Nr[r,\theta]\,Nr^{(1,0)}[r,\theta],\ \frac{N\theta[r,\theta]\,N\theta^{(0,1)}[r,\theta]+Nr[r,\theta]\left(N\theta[r,\theta]+r\,N\theta^{(1,0)}[r,\theta]\right)}{r^2}\right\}$$

In[30]:=
```
nDotGradn[NVec] - NVec Div[NVec] // Simplify
```

Out[30]=
$$\left\{-\frac{Nr[r,\theta]^2+N\theta[r,\theta]^2-N\theta[r,\theta]\,Nr^{(0,1)}[r,\theta]+Nr[r,\theta]\,N\theta^{(0,1)}[r,\theta]}{r},\ \frac{-N\theta[r,\theta]\,Nr^{(1,0)}[r,\theta]+Nr[r,\theta]\,N\theta^{(1,0)}[r,\theta]}{r}\right\}$$

- ### We now plug in the S = +- 1/2 defects

In[32]:=
```
NS = {Cos[θ] Cos[S θ] + Sin[θ] Sin[S θ], (-Sin[θ] Cos[S θ] + Cos[θ] Sin[S θ]) / r}
```

Out[32]=
$$\left\{\mathrm{Cos}[\theta]\,\mathrm{Cos}[S\,\theta]+\mathrm{Sin}[\theta]\,\mathrm{Sin}[S\,\theta],\ \frac{-\mathrm{Cos}[S\,\theta]\,\mathrm{Sin}[\theta]+\mathrm{Cos}[\theta]\,\mathrm{Sin}[S\,\theta]}{r}\right\}$$

In[33]:=
```
nDotGradn[NS] - NS Div[NS] // Simplify
```

Out[33]=
$$\left\{-\frac{S}{r},\ 0\right\}$$

- ### So the resulting field is radial, and the answer is...

In[34]:=
```
Integrate[(1 / 2) K24 (nDotGradn[NS] - NS Div[NS])[[1]] r, {θ, 0, 2 π}] // Simplify
```

Out[34]=
$$-K24\,\pi\,S$$